
    OGMate

An opinionated library for generating OG images for static and dynamic routes, mildly inspired by NimblePublisher.
OG, or Open Graph, images are shown by social media and chat apps as a preview image when sharing links, and although they don't do anything for SEO, they can make your links look more professional (or fun).
OGMate is designed specifically to reduce the lift to get OG images going for your Elixir blog or site, with an opinionated basic template. The library also provides escape hatches for setting up your own image generation and just using the boilerplate to organize things. It's agnostic to your actual routes and metadata, you provide the implementation for fetching image keys, titles, and descriptions.
For an end-to-end setup including a Plug, layout integration, and optional dev mode config, see the Getting started guide. Basic examples available below.
Installation
def deps do
  [
    {:og_mate, "~> 0.1.0"}
  ]
end
Examples
Happy path example.
defmodule MyApp.OGContent do
  def content_for("home"), do: {"MyApp", "Welcome."}
  def content_for("about"), do: {"About", "My cool site"}
  def content_for(_), do: :error
end

defmodule MyApp.OGImage do
  use OGMate,
    all_keys: ["home", "about"],
    content_for: MyApp.OGContent,
    theme: [
      background: "#0a0a0a",
      foreground: "#ffffff",
      font: "Inter",
      secondary: "#a3a3a3",
      logo: "priv/static/images/logo.png",
      site_name: "myapp.com"
    ],
    default: {"MyApp", "A brief site description."}
end

{:ok, png} = MyApp.OGImage.image_for("home")
MyApp.OGContent exports content_for/1 returning {title, description} for known keys or :error for unknown ones. Unknown keys resolve to the configured default image.
Or, replacing the template with your own renderer:
defmodule MyApp.OGRenderer do
  def render(title, description) do
    # Build your own image, return {:ok, png_binary} or {:error, reason}
    # Use the code from OGMate.Renderer as inspiration!
  end
end

defmodule MyApp.OGImage do
  use OGMate,
    all_keys: ["home", "about"],
    content_for: MyApp.OGContent,
    renderer: MyApp.OGRenderer,
    default: {"MyApp", "A brief site description."}
end
You can either set renderer or theme, you can't set both.
Theme
Required: background, foreground, font, secondary.
Optional: logo (path to image file), site_name (text shown next to logo).
See OGMate.Theme for the full spec.
Dev mode
By default, all images are built at compile time. For local dev, set dev_mode: true to render on demand instead of building everything up front:
use OGMate,
  all_keys: [...],
  content_for: MyApp.OGContent,
  theme: [...],
  default: {...},
  dev_mode: Application.compile_env(:my_app, :og_image_dev_mode, false)
In dev_mode, each image_for/1 call invokes content_for/1 and the renderer at runtime. New content in your content_for module is visible without expensive recompilations.
Dependencies
The default renderer uses :image, which bundles precompiled libvips binaries, no additional install needed. However, any fonts you want to use have to exist on the machine that builds the images.


  

    Getting started

TL;DR: define a content module mapping keys to titles and descriptions, point OGMate at it, mount a plug to serve /images/og/<key>.png, drop a meta tag in your layout. Images are built at compile time and served straight from memory.
OG images are those preview cards you see when sharing links on social media. They don't help SEO but they make your links look intentional. OGMate handles the build pipeline, you provide the content.
This walks through the full setup, framework-agnostic. Works inside a Phoenix endpoint, but Phoenix isn't required.
1. Add the dependency
def deps do
  [
    {:og_mate, "~> 0.1.0"}
  ]
end
2. Content module
Map URL slugs to {title, description}. Return :error for anything unknown, those fall through to the default image at runtime.
defmodule MyApp.OGContent do
  @static_content %{
    "home" => {"MyApp", "Welcome to MyApp."},
    "about" => {"About", "What MyApp does."}
  }

  def content_for(key) when is_map_key(@static_content, key),
    do: Map.fetch!(@static_content, key)

  def content_for("posts/" <> id) do
    case MyApp.Blog.find_by_id(id) do
      nil -> :error
      post -> {post.title, post.description}
    end
  end

  def content_for(_), do: :error
end
Static map for fixed pages, pattern-matched clauses for dynamic content. The catch-all :error clause at the bottom handles anything the rest didn't.
3. The OGMate module
defmodule MyApp.OGImage do
  use OGMate,
    all_keys:
      ["home", "about"] ++ Enum.map(MyApp.Blog.all_posts(), &"posts/#{&1.id}"),
    content_for: MyApp.OGContent,
    theme: [
      background: "#0a0a0a",
      foreground: "#ffffff",
      font: "Inter",
      secondary: "#a3a3a3",
      logo: "priv/static/images/logo.png",
      site_name: "myapp.com"
    ],
    default: {"MyApp", "Welcome to MyApp."},
    dev_mode: Application.compile_env(:my_app, :og_image_dev_mode, false)

  def path_for(key) when is_binary(key), do: "/images/og/#{key}.png"
end
all_keys is computed at compile time. MyApp.Blog.all_posts() is a normal call to an already-compiled module, the result gets inlined as a literal list. New posts means a recompile, but Phoenix code reloading handles that for you in dev.
path_for/1 keeps URL building in one place so the plug and your templates stay in sync.
4. The plug
A small plug intercepts /images/og/<key>.png and serves the bytes.
defmodule MyApp.Plugs.OGImage do
  @behaviour Plug
  import Plug.Conn

  @impl Plug
  def init(opts), do: opts

  @impl Plug
  def call(%Plug.Conn{request_path: "/images/og/" <> rest} = conn, _) do
    key = String.replace_suffix(rest, ".png", "")
    {:ok, bytes} = MyApp.OGImage.image_for(key)

    conn
    |> put_resp_content_type("image/png")
    |> put_resp_header("cache-control", "public, max-age=31536000")
    |> send_resp(200, bytes)
    |> halt()
  end

  def call(conn, _), do: conn
end
That cache-control is a year, which is fine because the URLs are content-keyed. If the content changes, the key changes.
5. Mount the plug
In Phoenix, drop it into the endpoint:
# lib/my_app_web/endpoint.ex
plug MyApp.Plugs.OGImage
plug Plug.Static, ...
Before Plug.Static so OG requests are caught first. Outside Phoenix, add it to your Plug.Router.
6. Reference in your HTML
<meta property="og:image" content={MyApp.OGImage.path_for(@og_key)} />
Set @og_key per route in your controller. "home" for the homepage, "posts/#{post.id}" for a post page.
7. Dev mode
Building everything at compile time is great for production. But rendering all your OG images on every save while you're iterating gets old fast. Set dev_mode: true and OGMate skips that at compile time, rendering lazily on each image_for/1 call instead. New content shows up immediately, no recompile.
# config/dev.exs
config :my_app, og_image_dev_mode: true
# config/prod.exs
config :my_app, og_image_dev_mode: false
The Application.compile_env(:my_app, :og_image_dev_mode, false) call in step 3 reads this at compile time.
See also
	OGMate: main module reference
	OGMate.Theme: theme field reference
	OGMate.Renderer: default renderer (1200×630 PNG layout)



  

    LICENSE


MIT License

Copyright (c) 2026 Johanna Larsson

Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all
copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE
SOFTWARE.

Builds PNG images at compile time from a user-provided key list and
content function. Always returns {:ok, bytes} from image_for/1.
Unknown keys resolve to the pre-built default image.
Example
defmodule MyApp.OGImage do
  use OGMate,
    all_keys: ["home", "about"],
    content_for: MyApp.OGContent,
    theme: [
      background: "#0a0a0a",
      foreground: "#ffffff",
      font: "Inter",
      secondary: "#a3a3a3",
      logo: "priv/static/images/logo.png",
      site_name: "myapp.com"
    ],
    default: {"MyApp", "A brief site description."}
end

{:ok, png} = MyApp.OGImage.image_for("home")
Custom Renderer
Pass a renderer: module that exports render(title, description):
defmodule MyApp.OGRenderer do
  def render(title, description), do: {:ok, png}
end

defmodule MyApp.OGImage do
  use OGMate,
    all_keys: [...],
    content_for: ...,
    renderer: MyApp.OGRenderer,
    default: {...}
end
:theme and :renderer are mutually exclusive.
See OGMate.Options for the full option contract, and the
Getting started guide for end-to-end setup.

Validates and normalizes options passed to use OGMate.
Options
	:all_keys (list of String.t/0) - Required.

	:content_for (atom/0) - Required.

	:default (tuple of String.t/0, String.t/0 values) - Required.

	:theme (keyword/0)
	:background (String.t/0) - Required.

	:foreground (String.t/0) - Required.

	:font (String.t/0) - Required.

	:secondary (String.t/0) - Required.

	:logo (String.t/0)

	:site_name (String.t/0)



	:renderer (atom/0)

	:dev_mode (boolean/0) - The default value is false.



      




  

    
OGMate.Renderer 
    



      
Default OG image renderer (1200×630).
Layout:
┌──────────────────────────────┐
│  [logo]  site_name           │  ← top-left, 80px padding
│                              │
│                              │
│                              │
│                              │
│                              │
│                              │
│  Title (72px bold)           │  ← bottom y = 470
│  Description (32px)          │  ← y = 500
└──────────────────────────────┘
Returns {:ok, png_binary} on success or {:error, reason} if any
rendering step fails (font unavailable, image read failure, etc.).

        render(title, description, theme)

      


        Render title + description + logo + site_name into a 1200×630 PNG.

          @spec render(String.t(), String.t(), OGMate.Theme.t()) ::
  {:ok, binary()} | {:error, term()}


      


Render title + description + logo + site_name into a 1200×630 PNG.
Returns {:ok, png_binary} on success, or {:error, reason} if any
rendering step fails.

Theme for OG image rendering.
Options
	:background (String.t/0) - Required.

	:foreground (String.t/0) - Required.

	:font (String.t/0) - Required.

	:secondary (String.t/0) - Required.

	:logo (String.t/0)

	:site_name (String.t/0)

        schema()

      


        Schema for theme keys. Exposed for embedding in other NimbleOptions schemas.



    


    
      
        validate!(theme)

      


        Validate a theme keyword list, return a %OGMate.Theme{} struct.

          @type t() :: %OGMate.Theme{
  background: String.t(),
  font: String.t(),
  foreground: String.t(),
  logo: String.t() | nil,
  secondary: String.t(),
  site_name: String.t() | nil
}

Schema for theme keys. Exposed for embedding in other NimbleOptions schemas.

          @spec validate!(keyword()) :: t()


      


Validate a theme keyword list, return a %OGMate.Theme{} struct.

